/*
 * Copyright (c) 2013, Juergen Bocklage-Ryannel, Johan Thelin
 * All rights reserved.
 *
 * Redistribution and use in source and binary forms, with or without
 * modification, are permitted provided that the following conditions are met:
 *     * Redistributions of source code must retain the above copyright
 *       notice, this list of conditions and the following disclaimer.
 *     * Redistributions in binary form must reproduce the above copyright
 *       notice, this list of conditions and the following disclaimer in the
 *       documentation and/or other materials provided with the distribution.
 *     * Neither the name of the editors nor the
 *       names of its contributors may be used to endorse or promote products
 *       derived from this software without specific prior written permission.
 *
 * THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS" AND
 * ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE IMPLIED
 * WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE ARE
 * DISCLAIMED. IN NO EVENT SHALL <COPYRIGHT HOLDER> BE LIABLE FOR ANY
 * DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR CONSEQUENTIAL DAMAGES
 * (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF SUBSTITUTE GOODS OR SERVICES;
 * LOSS OF USE, DATA, OR PROFITS; OR BUSINESS INTERRUPTION) HOWEVER CAUSED AND
 * ON ANY THEORY OF LIABILITY, WHETHER IN CONTRACT, STRICT LIABILITY, OR TORT
 * (INCLUDING NEGLIGENCE OR OTHERWISE) ARISING IN ANY WAY OUT OF THE USE OF THIS
 * SOFTWARE, EVEN IF ADVISED OF THE POSSIBILITY OF SUCH DAMAGE.
 */

// showcase.qml

import QtQuick 2.5
import QtGraphicalEffects 1.0

Image {
    // 根元素id习惯命名为root
    id: root
    source: "images/background.png"

    // 自定义属性
    property int blurRadius: 0
    // 属性绑定：将testPropertyBind的值绑定为blurRadius + 10
    // 属性赋值会使属性绑定失效，这里测试空格键抬起后testPropertyBind始终为100，不会随blurRadius变化了
    property int testPropertyBind : blurRadius + 10

    Image {
        id: pole
        anchors.horizontalCenter: parent.horizontalCenter
        anchors.bottom: parent.bottom
        source: "images/pole.png"
    }


    Image {
        id: wheel
        anchors.centerIn: parent
        source: "images/pinwheel.png"
        Behavior on rotation {
            NumberAnimation {
                duration: 250
            }
        }
        // 信号，当rotation改变时触发，此时我们向控制台打印输出
        onRotationChanged: console.log('rotation:', rotation)
        layer.effect: FastBlur {
            id: blur
            radius: root.blurRadius
            Behavior on radius {
                NumberAnimation {
                    duration: 250
                }
            }
        }
        layer.enabled: true
    }

    MouseArea {
        anchors.fill: parent
        onPressed: {
            wheel.rotation += 90
            root.blurRadius = 16
        }
        onReleased: {
            root.blurRadius = 0
        }
    }

    // 处理键盘事件需要focus
    focus: true
    // 空格按下
    Keys.onSpacePressed: {
        console.log('onSpacePressed')
        // 执行js函数
        increment()
    }

    Keys.onEnterPressed: {
        console.log('testPropertyBind', testPropertyBind)
    }

    // 按键抬起
    Keys.onReleased: {
        // 空格
        if (event.key == Qt.Key_Space) {
            console.log('onSpaceReleased')
            root.blurRadius = 0

            // 赋值破坏了属性绑定，以后testPropertyBind不会随blurRadius变化了
            root.testPropertyBind = 100
        }
    }

    // 定义js函数
    function increment() {
        wheel.rotation += 90
        root.blurRadius = 16
    }
}
